
// https://www.cepstrum.co.jp/

//         

//                        +---+ d             

//                   +--->| h +---+     

//                   |    +---+   |      

//                   |          + v  e    

// white noise  x ---+      ^    (+)---+--->

//                   |      |   - ^    |   

//                   |    +---+   |    |

//                   +--->| w +---+    |

//                        +---+ y      |

//                          |          |

//                          +----------+



clear;                   // clear all variables



rand('normal');



DATLEN=200000;           // data length 

FS    =8000;             // sampling frequency [Hz]

FIRLEN=1024;             // adaptive filter legnth

GAIN  =4.0;              // h() gain

MU    =0.0003;           // step size parameter of adaptive filter



indata=rand(1:DATLEN);                  // generate white noise

indata=indata/max(abs(indata));         // scaling 

outdata=zeros(1:DATLEN);



xbuf=zeros(1:FIRLEN);

w=zeros(1:FIRLEN);



// BPF coefficient, Butterworh 250Hz/3900Hz (fs=8kHz)

A = 8.233127742944482685D-01;

B1=-1.652569066320155322D-01;

B2= 1.603848922725006343D+00;

B3= 1.115318355518675197D-01;

B4=-6.779093254568269167D-01;



// generate impulse response h()

rand('seed', 3141);

h=[zeros(1:5), rand(1:FIRLEN-5)];

h=h.*(0.975^(1:FIRLEN)); 

h=h/max(abs(fft(h, -1)));

h=GAIN*h;



// apply BPF to h(), Butterworth 250Hz/3900Hz

zx=0; zzx=0; zzzx=0; zzzzx=0; 

zy=0; zzy=0; zzzy=0; zzzzy=0;

for i=1:FIRLEN

  x=h(i);

  y=A*(x-2.0*zzx+zzzzx)+B1*zy+B2*zzy+B3*zzzy+B4*zzzzy;

  zzzzx=zzzx; zzzx=zzx; zzx=zx; zx=x; 

  zzzzy=zzzy; zzzy=zzy; zzy=zy; zy=y;

  h(i)=y;

end



scf(0);

clf;

plot(h);        // plot h()



// plot h() frequency response

scf(1);

clf;

s=20.0*log10(abs(fft(h, -1))+1e-99);

s=s(1:FIRLEN/2);

plot2d((FS/2.0)*(0:(FIRLEN/2-1))/(FIRLEN/2), s, axesflag=1, rect=[0, -60, FS/2, 20], style=1);

xlabel "frequency [Hz]"

ylabel "[dB]"



for i=1:DATLEN 

  x=indata(i);



  xbuf=[x, xbuf(1:FIRLEN-1)];

  d=h*xbuf';

  y=w*xbuf';

  e=d-y;

  w=w+2.0*MU*e*xbuf;       // LMS algorithm



  outdata(i)=e;

end



scf(2);

clf;

plot(outdata);          // plot e



// plot w() frequency response

s=20.0*log10(abs(fft(w, -1))+1e-99);

s=s(1:FIRLEN/2);

scf(3);

clf;

plot2d((FS/2.0)*(0:(FIRLEN/2-1))/(FIRLEN/2), s, axesflag=1, rect=[0, -60, FS/2, 20], style=2);

xlabel "frequency [Hz]"

ylabel "[dB]"




// https://www.cepstrum.co.jp/

//         

//                         +---+ d             

//                    +--->| h +---+     

//                    |    +---+   |      

//                    |          + v  e    

// band limited  x ---+      ^    (+)---+--->

// white noise        |      |   - ^    |   

//                    |    +---+   |    |

//                    +--->| w +---+    |

//                         +---+ y      |

//                           |          |

//                           +----------+



clear;                   // clear all variables



rand('normal');



DATLEN=200000;           // data file length 

FS    =8000;             // sampling frequency [Hz]

FIRLEN=1024;             // adaptive filter legnth

GAIN  =4.0;              // h() gain

MU    =0.0001;           // step size parameter of adaptive filter



indata=loadwave('bp_noise.wav');     // read band limited white noise wav file

outdata=zeros(1:DATLEN);



xbuf=zeros(1:FIRLEN);

w=zeros(1:FIRLEN);



// BPF Butterworh, 250Hz/3900Hz (fs=8kHz)

A = 8.233127742944482685D-01;

B1=-1.652569066320155322D-01;

B2= 1.603848922725006343D+00;

B3= 1.115318355518675197D-01;

B4=-6.779093254568269167D-01;



// generate impulse response of acoustice system (h)

rand('seed', 3141);

h=[zeros(1:5), rand(1:FIRLEN-5)];

h=h.*(0.975^(1:FIRLEN)); 

h=h/max(abs(fft(h, -1)));

h=GAIN*h;



// apply BPF to h(), Butterworth 250Hz/3900Hz

zx=0; zzx=0; zzzx=0; zzzzx=0; 

zy=0; zzy=0; zzzy=0; zzzzy=0;

for i=1:FIRLEN

  x=h(i);

  y=A*(x-2.0*zzx+zzzzx)+B1*zy+B2*zzy+B3*zzzy+B4*zzzzy;

  zzzzx=zzzx; zzzx=zzx; zzx=zx; zx=x; 

  zzzzy=zzzy; zzzy=zzy; zzy=zy; zy=y;

  h(i)=y;

end



scf(0);

clf;

plot(h);        // plot h()



// plot h() frequency response

scf(1);

clf;

s=20.0*log10(abs(fft(h, -1))+1e-99);

s=s(1:FIRLEN/2);

plot2d((FS/2.0)*(0:(FIRLEN/2-1))/(FIRLEN/2), s, axesflag=1, rect=[0, -60, FS/2, 20], style=1);

xlabel "frequency [Hz]"

ylabel "[dB]"



for i=1:DATLEN 

  x=indata(i);



  xbuf=[x, xbuf(1:FIRLEN-1)];

  d=h*xbuf';

  y=w*xbuf';

  e=d-y;

  w=w+2.0*MU*e*xbuf;       // LMS algorithm



  outdata(i)=e;

end



scf(2);

clf;

plot(outdata);          // plot e



// plot w() frequency response

s=20.0*log10(abs(fft(w, -1))+1e-99);

s=s(1:FIRLEN/2);

scf(3);

clf;

plot2d((FS/2.0)*(0:(FIRLEN/2-1))/(FIRLEN/2), s, axesflag=1, rect=[0, -60, FS/2, 20], style=5);

xlabel "frequency [Hz]"

ylabel "[dB]"



